
								I have encountered a problem when using the fcl-pdf package
with regards to different behavior with Unicode characters.

The cli test application generates a correct PDF:
Code: Pascal  [Select][+][-]
	program pdfcli;

	{$mode objfpc}{$H+}

	uses

	  classes,

	  sysutils,

	  fppdf,

	  ShellApi;

	 

	var

	 FDoc: TPDFDocument;

	 FPage: TPDFPage;

	 FtBold: Integer;

	 Section: TPDFSection;

	 F: TFileStream;

	 FileName : String;

	 

	begin

	  FDoc := TPDFDocument.Create(Nil);

	  try

	    FDoc.Options := [poPageOriginAtTop, poUseImageTransparency, poMetadataEntry];

	    FDoc.StartDocument;

	    Section := FDoc.Sections.AddSection;

	    FPage := FDoc.Pages.AddPage;

	    FPage.PaperType := ptA4;

	    FPage.UnitOfMeasure := uomMillimeters;

	    Section.AddPage(FPage);

	    FtBold := FDoc.AddFont('Helvetica-Bold');

	 

	    FPage.SetFont(FtBold, 16);

	    FPage.SetColor(clBlack, False);

	    FPage.WriteText(150, 150, 'æøå');

	 

	    FileName := GetTempFileName('', 'invoice');

	    FileName := FileName + '.pdf';

	 

	    F := TFileStream.Create(FileName,fmCreate);

	    try

	      FDoc.SaveToStream(F);

	    finally

	      F.Free;

	    end;

	 

	    ShellExecute(0, 'open', PChar(FileName), Nil, PChar(ExtractFilePath(FileName)),1);

	  finally

	    FreeAndNil(FDoc);

	  end;

	end.

	 




The same code as a gui application generates garbage:
Code: Pascal  [Select][+][-]
	program pdfgui;

	{$mode delphi}{$H+}

	{$codepage UTF8}

	uses

	  {$IFDEF UNIX}{$IFDEF UseCThreads}

	  cthreads,

	  {$ENDIF}{$ENDIF}

	  Classes, SysUtils, Interfaces, Forms, StdCtrls,

	  fppdf, ShellApi;

	 

	type

	  TMyForm = class(TForm)

	  public

	    MyButton: TButton;

	    procedure ButtonClick(ASender: TObject);

	    constructor Create(AOwner: TComponent); override;

	  end;

	 

	procedure TMyForm.ButtonClick(ASender:TObject);

	var

	 FDoc: TPDFDocument;

	 FPage: TPDFPage;

	 FtBold: Integer;

	 Section: TPDFSection;

	 F: TFileStream;

	 FileName : String;

	begin

	  FDoc := TPDFDocument.Create(Nil);

	  try

	    FDoc.Options := [poPageOriginAtTop, poUseImageTransparency, poMetadataEntry];

	    FDoc.StartDocument;

	    Section := FDoc.Sections.AddSection;

	    FPage := FDoc.Pages.AddPage;

	    FPage.PaperType := ptA4;

	    FPage.UnitOfMeasure := uomMillimeters;

	    Section.AddPage(FPage);

	    FtBold := FDoc.AddFont('Helvetica-Bold');

	 

	    FPage.SetFont(FtBold, 16);

	    FPage.SetColor(clBlack, False);

	    FPage.WriteText(150, 150, 'æøå');

	 

	    FileName := GetTempFileName('', 'invoice');

	    FileName := FileName + '.pdf';

	 

	    F := TFileStream.Create(FileName,fmCreate);

	    try

	      FDoc.SaveToStream(F);

	    finally

	      F.Free;

	    end;

	 

	    ShellExecute(0, 'open', PChar(FileName), Nil, PChar(ExtractFilePath(FileName)),1);

	  finally

	    FreeAndNil(FDoc);

	  end;

	 

	  Close;

	end;

	 

	constructor TMyForm.Create(AOwner: TComponent);

	begin

	  inherited Create(AOwner);

	  Position := poScreenCenter;

	  Height := 400;

	  Width := 400;

	 

	  VertScrollBar.Visible := False;

	  HorzScrollBar.Visible := False;

	 

	  MyButton := TButton.Create(Self);

	  with MyButton do

	  begin

	    Height := 30;

	    Left := 100;

	    Top := 100;

	    Width := 100;

	    Caption := 'Close';

	    OnClick := ButtonClick;

	    Parent := Self;

	  end;

	end;

	 

	var

	  MyForm : TMyForm;

	 

	begin

	  Application.Scaled:=True;

	  Application.Initialize;

	  Application.CreateForm(TMyForm, MyForm);

	  Application.Run;

	end.

	 




I see the fcl-pdf package used the UTF8String in the WriteText procedure:
Code: Pascal  [Select][+][-]
	Procedure WriteText(X, Y: TPDFFloat; AText : UTF8String; const ADegrees: single = 0.0; const AUnderline: boolean = false; const AStrikethrough: boolean = false); overload;




Is this creating the problems?

I have tried inserting various converting routines available, but without any luck.

The test is performed on Windows 10 with Lazarus IDE v.2.0.10 r63526.

								In the second example you use {$codepage UTF8}, but not in the first. Read String Literals.

This doesn't work for me anyway, because fcl-pdf for standard fonts only assumes ansi characters, and for other characters, fonts need to be embedded. For example, to make it work, needed to do this:
Code: Pascal  [Select][+][-]
	FDoc.AddFont('c:\Windows\Fonts\segoeui.ttf', 'Segoe UI');

								Looking at the generated PDF file with the CLI application:
Code: Pascal  [Select][+][-]
	stream

	/F0 16 Tf

	0 0 0 rg

	BT

	425.20 416.80 TD

	(Ã¦Ã¸Ã¥) Tj

	ET

	 

	endstream

	endobj

	 

	8 0 obj

	<<

	/Type /Font

	/Subtype /Type1

	/Encoding /WinAnsiEncoding

	/FirstChar 32

	/LastChar 255

	/BaseFont /Helvetica-Bold

	/Name /F0

	>>

	endobj

	 




The content looks to be written in ANSI/ : (Ã¦Ã¸Ã¥)

With the GUI code:
Code: Pascal  [Select][+][-]
	stream

	/F0 16 Tf

	0 0 0 rg

	BT

	425.20 416.80 TD

	(XE6XF8XE5) Tj

	ET

	 

	endstream

	endobj

	 

	8 0 obj

	<<

	/Type /Font

	/Subtype /Type1

	/Encoding /WinAnsiEncoding

	/FirstChar 32

	/LastChar 255

	/BaseFont /Helvetica-Bold

	/Name /F0

	>>

	endobj

	 




The content looks to be written in UTF8 : (I recognize the sequence XE6XF8XE5)

This is probably the reason the CLI application works, but 
the GUI application fails.  The CLI application writes the file
in the windows codepage CP1252 and the GUI application
write the file in UTF8.

The font then fails to render correctly for the UTF8 content
as it is has WinAnsiEncoding.

I expected something in the lines of:
Code: Pascal  [Select][+][-]
	FPage.WriteText(150, 150, RawBytestring(UTF8ToCP1252('æøå')));

	 



to work, but is does not. (I can hand edit the generated PDF file
with the GUI, change codepage to CP1252 and  type the character.
The PDF then renders correctly). 

I will look into adding proper Unicode fonts to the application.

Thanks for the help.

								I got it to work with the following code:
Code: Pascal  [Select][+][-]
	procedure TMyForm.ButtonClick(ASender:TObject);

	var

	 FDoc: TPDFDocument;

	 FPage: TPDFPage;

	 FtBold: Integer;

	 Section: TPDFSection;

	 F: TFileStream;

	 FM : TMemoryStream;

	 FileName: String;

	 SR : RawByteString;

	 SA : AnsiString;

	begin

	  FDoc := TPDFDocument.Create(Nil);

	  try

	    FDoc.Options := [poPageOriginAtTop, poUseImageTransparency, poMetadataEntry];

	    FDoc.StartDocument;

	    Section := FDoc.Sections.AddSection;

	    FPage := FDoc.Pages.AddPage;

	    FPage.PaperType := ptA4;

	    FPage.UnitOfMeasure := uomMillimeters;

	    Section.AddPage(FPage);

	    FtBold := FDoc.AddFont('Helvetica-Bold');

	 

	    FPage.SetFont(FtBold, 16);

	    FPage.SetColor(clBlack, False);

	 

	    FPage.WriteText(150, 150, 'æøå');

	 

	    FileName := GetTempFileName('', 'invoice');

	    FileName := FileName + '.pdf';

	 

	    F := TFileStream.Create(FileName,fmCreate);

	    FM := TMemoryStream.Create();

	    try

	      FDoc.SaveToStream(FM);

	      SetString(SA, PAnsiChar(FM.Memory), FM.Size);

	      System.SetCodePage(RawByteString(SA), CP_UTF8 ,True);

	      SR := UTF8ToISO_8859_1(SA);

	      F.Write(SR[1], Length(SR));

	    finally

	      FM.Free;

	      F.Free;

	    end;

	 

	    ShellExecute(0, 'open', PChar(FileName), Nil, PChar(ExtractFilePath(FileName)),1);

	  finally

	    FreeAndNil(FDoc);

	  end;

	  Close;s

	end;




I read the content of the PDF to string and use the function
UTF8ToISO_8859_1 to convert to a ANSI codepage which
is written to disk.

I assume this solution is OK as I expect the standard PDF fonts to
be the same everywhere.

								Researching the subject a bit is seems the standard PDF fonts
has their own encoding, called: StandardEncoding, MacRomanEncoding,
WinAnsiEncoding and PDFDocEncoding.

The WinAnsiEncoding intersect partly with the CP1252 and ISO-8859-1
code pages.

The fcl-pdf package should write text content using the standard
fonts in ANSI and perform a check if the UTF8 codepoint can be
translated to the PDF standard encoding without loss of information.

This is probably involved as any of the standard code pages can
not be used directly if the standard is to be followed.

								@tennis: thank you so much for your research and providing a solution! This saved me hours 

for anyone wondering where to find the function UTF8ToISO_8859_1 - it's defined in LConvEncoding

